
/*Supplemental material : SAS codes for analyzing linkage mapping populations 

for marker main- and marker × treatment (environment) interactions as well as 

epistasis (marker × marker) and epistasis × treatment (environment) 

interaction. In following codes, the word “treatment” represents 

“environment”.  */ 

 

/* Create the simulated data files and install the marcos  */  

 

 

/****************************************************************************/ 

           /*Declare variables used to define simulation scenarios */ 

/****************************************************************************/ 

 

%let No_Genotypes=250;         /*Specify the number of tested 

genotypes in the estimation set*/ 

 

%let No_SNP=5;                         /*Specify the number of all SNP markers 

to simulate*/ 

 

/****************************************************************************/ 

                           /*Simulation of datasets*/ 

/****************************************************************************/ 

 

/*Simulate vector u of dimension p (No of SNP) of marker effects*/ 

 

/********* See also supplementary material Piepho HP, Ogutu JO, Schulz-Streeck 

T, Estaghvirou B, Gordillo A, Technow F. 2012. Efficient Computation of Ridge-

Regression Best Linear Unbiased Prediction in Genomic Selection in Plant 

Breeding. Crop Science 52, 1093–1104. ***/ 

/******************** for the simulation of the test data 

***************************************************************/ 

 

 

data u; 

array u u1-u&No_SNP; 

do i=1 to &No_SNP; 

  u[i]=normal(59752)*sqrt(1/&No_SNP);                           

/*var(u)=var(e)/No_SNP*/ 

end; 

output; 

run; 

 

/*Simulate genotypic values (g) and phenotypic values (y =genotypic 

effect+random effect) for the estimation 

set, i.e. part of the data set with only the n tested genotypes*/ 

 

data g; 

set u; 

array SNP SNP1-SNP&No_SNP; 

array u u1-u&No_SNP; 

do j=1 to &No_Genotypes; 

  g=0; 

  do i=1 to &No_SNP; 

    x=ranbin(69752,1,0.5); 

    SNP[i]=2*x-1; 

    g=g+SNP[i]*u[i]; 

    y=g+normal(174265); 

  end; 



  keep  SNP1-SNP&No_SNP g y ; 

  output; 

end; 

run; 

Data g; set g;  

 Genotype=_N_;  

Run; 

data g_treat; 

set u; 

array SNP SNP1-SNP&No_SNP; 

array u u1-u&No_SNP; 

do j=1 to &No_Genotypes; 

  g=0; 

  do i=1 to &No_SNP; 

    x=ranbin(123456,1,0.5); 

    SNP[i]=2*x-1; 

    g=g+SNP[i]*u[i]; 

    y=g+normal(987654); 

  end; 

  keep  SNP1-SNP&No_SNP g y ; 

  output; 

end; 

run; 

Data g_treat; set g_treat;  

 Genotype=_N_; Treatment=1; 

Run; 

Data g_treat_1; set g;  

 Genotype=_N_; Treatment=2; 

Run; 

Proc append Base=g_treat data=g_treat_1 force; run ;  

proc delete data = work.g_treat_1; run; 

proc delete data = work.u; run; 

 

 

 /***********************************************************************

*/ 

 

     /* Macros for analyzing Marker-trait 

associations */ 

 

 

 /***********************************************************************

*/ 

 

%macro nobs(ds); 

 DATA _NULL_; 

  IF _N_ = 0 THEN DO; 

   I = 1; 

   SET &ds POINT=I NOBS=NOBS; 

  END; 

  CALL SYMPUT('NOBS',TRIM(LEFT(PUT(NOBS,6.)))); 

  STOP; 

 RUN; 

%mend nobs; 

 

 

%macro marker_main_effect; 

 



 proc delete data = work.Marker_Main_effect; run; 

 

 %do j=1 %to   &No_SNP  ;  

 

  %let marker=marker_&j; 

 

  /*********** Hide log information **************************/ 

  %if &j gt 3  %then %do; 

    proc printto log=outlog;     run; 

    proc options;      run; 

    proc printto print = outdoc;  run;  

    ODS Results OFF;  

   %end; 

  /*************************************************************/ 

 

  /************** Model; change for your experimental design ***/ 

 

 

  proc mixed data=G;  

   class genotype SNP&j; 

   model y= SNP&j / htype=1 solution; 

   ods output   tests1=SNP_test_Anova; 

  run; 

 

 

  /*************************************************************/ 

 

 

  Data SNP_test_Anova; set SNP_test_Anova;   

   informat   effect $25.  Marker $25.  probf 30.28; 

   format   effect $25.  Marker $25.  probf 30.28;  

   Marker="&Marker";  

  run; 

  %if &j eq 1 %then %do; 

   Data Marker_Main_effect; set SNP_test_Anova;  run; 

  %end; 

  %else %do; 

   Proc append Base=Marker_Main_effect data=SNP_test_Anova 

force; run ;  

  %end; 

 %end; 

 Proc sort data=Marker_Main_effect; by ProbF; run; 

 proc delete data = work.SNP_test_Anova; run; 

 

 /*********** Calculation of FDR probability ******************/ 

 

 proc multtest inpvalues(ProbF)=Marker_Main_effect OUT=Marker_Main_effect 

FDR; run; 

 

 /*************************************************************/ 

 

 /**************** Print Log file information again ************/ 

 

 Proc printto; run;  

 

 /*************************************************************/ 

 

%Mend marker_main_effect; 



 

 

%macro marker_Treatment_interaction; 

 proc delete data = work.Marker_treat_interaction; run; 

 proc delete data = work.Marker_main_effect_in_treat; run; 

 

 %do j=1 %to   &No_SNP  ;  

 

  %let marker=marker_&j; 

 

  /*********** Hide log information **************************/ 

  %if &j gt 3  %then %do; 

    proc printto log=outlog;     run; 

    proc options;      run; 

    proc printto print = outdoc;  run;  

    ODS Results OFF;  

   %end; 

  /*************************************************************/ 

 

 

 

  /************** Model; change according to your experimental 

design ***/ 

 

 

   proc mixed data=g_treat;  

   class genotype treatment SNP&j; 

   model y= treatment SNP&j treatment*SNP&j / htype=1 solution; 

   ods output   tests1=SNP_by_treat_test_Anova; 

   run; 

 

  /*************************************************************/ 

 

  Data SNP_by_treat_test_Anova;  set SNP_by_treat_test_Anova;  

   informat  effect $25.  Marker $25.  probf 30.28; 

   format   effect $32.  Marker $25.  probf 30.28;  

   Marker="&Marker";  

  run; 

  Data SNP_treat_interaction;  set SNP_by_treat_test_Anova;  If 

effect eq  "Treatment*SNP&j"; run; 

  Data SNP_effect;     set SNP_by_treat_test_Anova;  

If effect eq  "SNP&j"; run; 

  %if &j le 2 %then %do; 

   Data Marker_treat_interaction; set SNP_treat_interaction;  

run; 

   Data Marker_main_effect_in_treat; set SNP_effect;  run; 

  %end; 

  %else %do; 

   Proc append Base=Marker_treat_interaction 

 data=SNP_treat_interaction force; run ;  

   Proc append Base=Marker_main_effect_in_treat  

 data=SNP_effect force; run ;  

  %end; 

 %end; 

 

 /*********** Calculation of FDR probability ******************/ 

 



 proc multtest inpvalues(ProbF)=Marker_treat_interaction 

OUT=Marker_treat_interaction FDR; run; 

 proc multtest inpvalues(ProbF)=Marker_main_effect_in_treat 

OUT=Marker_main_effect_in_treat FDR; run; 

 

 /*************************************************************/ 

 

 Data Marker_main_effect_in_treat; set Marker_main_effect_in_treat; 

Main_NumDF=NumDF; Main_DenDF=DenDF; Main_FValue=FValue; Main_ProbF=ProbF; 

Main_FDR_p=fdr_p;  

  Drop NumDF DenDF FValue  ProbF fdr_p; 

 run; 

 Data Marker_treat_interaction; set Marker_treat_interaction; 

Treat_NumDF=NumDF; Treat_DenDF=DenDF; Treat_FValue=FValue; Treat_ProbF=ProbF; 

Treat_FDR_p=fdr_p;  

  Drop NumDF DenDF FValue  ProbF fdr_p; 

 run; 

 Proc sort data=Marker_treat_interaction;  by Marker; run; 

 Proc sort data=Marker_main_effect_in_treat; by Marker; run; 

 data Marker_treat_interaction; merge Marker_main_effect_in_treat 

Marker_treat_interaction; by Marker; run; 

 

 Proc sort data=Marker_treat_interaction;  by Treat_ProbF; run; 

 

 proc delete data = work.SNP_treat_interaction; run; 

 proc delete data = work.SNP_effect; run; 

 proc delete data = work.SNP_by_treat_test_Anova; run; 

 proc delete data = work.Marker_main_effect_in_treat; run; 

 

 /**************** Print Log file information again ************/ 

 

 Proc printto; run;  

 

 /*************************************************************/ 

 

%Mend marker_Treatment_interaction; 

 

 

%macro Epistatic_effect; 

 

 

  /*********** Hide results as the system would be slow otherwise 

**************************/ 

 

   ods graphics off; 

   ods html close; 

  

 

 /***********++++++++++++++++++++++++++++++++++++++++++++++++++++********

******************/ 

 

 proc delete data = work.Epistatic_effect; run; 

 

 %do i=1 %to &No_SNP -1  ; 

  %do j=&i+1 %to  &No_SNP  ; 

   %let marker_a=marker_&i; %let marker_b=marker_&j;  

 

  /*********** Hide log information **************************/ 



   %if &marker_a eq marker_1  %then %do; 

    %if &marker_b eq marker_3 %then %do; 

     proc printto log=outlog;     run; 

     proc options;      run; 

     proc printto print = outdoc;  run;  

     ODS Results OFF;  

    %end; 

   %end; 

  /*************************************************************/ 

 

   Data SNP_test_Anova;  

    informat  Effect $32.  Marker_a $25. Marker_b $25. 

probf 30.28; 

    format   Effect $32.  Marker_a $25. Marker_b 

$25. probf 30.28; 

    length   Effect $32; 

   run; 

 

  /************** Model; change according to your experimental 

design ***/ 

 

   proc mixed data=G;  

    class genotype SNP&i SNP&j; 

    model y= SNP&i*SNP&j / htype=1 solution; 

    ods output   tests1=SNP_test_Anova; 

   run; 

 

 

 

 /**********************************************************************/ 

 

   Data SNP_test_Anova; set SNP_test_Anova;  

    informat  Effect $32.  Marker_a $25. Marker_b $25. 

probf 30.28; 

    format   Effect $32.  Marker_a $25. Marker_b 

$25. probf 30.28; 

    length   Effect $32; 

    Marker_a="&marker_a"; Marker_b="&marker_b";  

   run; 

   %if &i eq 1 and &j eq 2 %then %do; 

    Data Epistatic_effect; set SNP_test_Anova;  run; 

   %end; 

   %else %do; 

    Proc append Base=SNP_test_Anova data=Epistatic_effect 

force; run ;  

    Data Epistatic_effect; set SNP_test_Anova; run; 

   %end; 

  %end; 

 %end; 

 ODS Results on; ods html; 

 Proc sort data=Epistatic_effect; by ProbF; run; 

 proc delete data = work.SNP_test_Anova; run; 

 

 /*********** Calculation of FDR probability ******************/ 

 

 proc multtest inpvalues(ProbF)=Epistatic_effect OUT=Epistatic_effect 

FDR; run; 

 



 /*************************************************************/ 

 

 /**************** Print Log file information again ************/ 

 

 Proc printto; run; 

 

 /*************************************************************/ 

 

%Mend Epistatic_effect; 

 

 

%macro Epistatic_Treatment_interaction; 

 

  /*********** Hide results as the system would be slow otherwise 

**************************/ 

 

   ods graphics off; 

   ods html close; 

  

 

 /***********++++++++++++++++++++++++++++++++++++++++++++++++++++********

******************/ 

 

 proc delete data = work.Epistatic_treat_interaction; run; 

 proc delete data = work.Epistatic_effect_in_treat; run; 

 

 %do i=1 %to &No_SNP -1  ; 

  %do j=&i+1 %to  &No_SNP  ; 

   %let marker_a=marker_&i; %let marker_b=marker_&j;  

 

   /*********** Hide log information 

**************************/ 

   %if &marker_a eq marker_1  %then %do; 

    %if &marker_b eq marker_3 %then %do; 

     proc printto log=outlog;     run; 

     proc options;      run; 

     proc printto print = outdoc;  run;  

     ODS Results OFF;  

    %end; 

   %end; 

  

 /*************************************************************/ 

 

  /************** Model; change according to your experimental 

design ***/ 

 

 

   proc mixed data=G_treat;  

    class genotype Treatment SNP&i SNP&j; 

    model y= Treatment SNP&i*SNP&j Treatment*SNP&i*SNP&j/ 

htype=1 solution; 

    ods output   tests1=SNP_by_treat_test_Anova; 

   run; 

 

  /*************************************************************/ 

 

   Data SNP_by_treat_test_Anova; set SNP_by_treat_test_Anova;  



    informat  Effect $32.   Marker_a $25. Marker_b $25. 

probf 30.28; 

 

    format   Effect $32.   Marker_a $25. Marker_b 

$25. probf 30.28; 

    Marker_a="&marker_a"; Marker_b="&marker_b";  

   run; 

   Data Epi_effect;   set SNP_by_treat_test_Anova; where 

effect contains  "SNP&i"; run; 

   Data Epi_effect_treat;  set Epi_effect; where effect contains  

"Treatment"; run; 

   Data Epi_effect;   set Epi_effect; where effect not 

contains  "Treatment"; run; 

   %if i eq 1 and j eq 2 %then %do; 

    Data Epistatic_treat_interaction; set 

Epi_effect_treat;  run; 

    Data Epistatic_effect_in_treat; set Epi_effect;  run; 

   %end; 

   %else %do; 

    Proc append Base=Epi_effect_treat 

data=Epistatic_treat_interaction force; run ;  

    Proc append Base=Epi_effect  

data=Epistatic_effect_in_treat force; run ;  

    Data Epistatic_treat_interaction; set 

Epi_effect_treat; run; 

    Data Epistatic_effect_in_treat; set Epi_effect; run; 

   %end; 

  %end; 

 %end; 

 ODS Results on; ods html; 

 

 

 /*********** Calculation of FDR probability ******************/ 

 

 proc multtest inpvalues(ProbF)=Epistatic_treat_interaction 

OUT=Epistatic_treat_interaction FDR; run; 

 proc multtest inpvalues(ProbF)=Epistatic_effect_in_treat 

OUT=Epistatic_effect_in_treat FDR; run; 

 

 /*************************************************************/ 

 

 Data Epistatic_effect_in_treat; set Epistatic_effect_in_treat; 

Epi_NumDF=NumDF; Epi_DenDF=DenDF; Epi_FValue=FValue; Epi_ProbF=ProbF; 

Epi_FDR_p=fdr_p;  

  Drop NumDF DenDF FValue  ProbF fdr_p; 

 run; 

 Data Epistatic_treat_interaction; set Epistatic_treat_interaction; 

Epi_Treat_NumDF=NumDF; Epi_Treat_DenDF=DenDF; Epi_Treat_FValue=FValue; 

Epi_Treat_ProbF=ProbF; Epi_Treat_FDR_p=fdr_p;  

  Drop NumDF DenDF FValue  ProbF fdr_p; 

 run; 

 Proc sort data=Epistatic_effect_in_treat;  by marker_a marker_b; 

run; 

 Proc sort data=Epistatic_treat_interaction; by marker_a marker_b; run; 

 data Epistatic_treat_interaction; merge Epistatic_effect_in_treat 

Epistatic_treat_interaction; by marker_a marker_b; run; 

 

 Proc sort data=Epistatic_treat_interaction;  by Epi_Treat_ProbF; run; 



 

 proc delete data = work.Epistatic_effect_in_treat; run; 

 proc delete data = work.SNP_by_treat_test_Anova; run; 

 proc delete data = work.Epi_effect; run; 

 proc delete data = work.Epi_effect_treat; run; 

 

 /**************** Print Log file information again ************/ 

 

 Proc printto; run; 

 

 /*************************************************************/ 

 

%Mend Epistatic_Treatment_interaction; 

 

 

/*********************************************************************** 

 

Start the macro by calling it using %(the given) macro: 

 

Macro for Marker-trait associations/Epistasis use data corrected by population 

structure and/or kinship:  

 

%marker_main_effect; 

 

%Epistatic_effect; 

 

Macro for Marker-trait associations/Epistasis including treatment 

(environment: fixed factor) interactions use data corrected by population 

structure and/or kinship:  

 

%marker_Treatment_interaction; 

 

%Epistatic_Treatment_interaction; 

 

Macro for Marker-trait associations/Epistasis including treatment calculates 

the main effect and the treatment (environment) interaction effect as the same 

time  

and results are presented in the same table e. g. Main_ProbF and Treat_ProbF 

or Epi_ProbF and Epi_Treat_ProbF.   

 

 

************************************************************************/ 

 

 

 

 

%Macro cross_validation; 

  %let randomini=123345; 

  %do cv=1 %to &No_CV_reps  ; 

   %let marker=marker_&j; 

 

   data Crossvalidation_dataset Crossvalidation_dataset_1;  set 

g; 

    Keep SNP&j y genotype;  

   run; 

   Proc sort data=Crossvalidation_dataset_1; by genotype; run;  

    data Crossvalidation_dataset_1; set 

Crossvalidation_dataset_1; by genotype; 



    if first.genotype;  

   run; 

   Proc sort data=Crossvalidation_dataset_1; by SNP&j; run;  

    data Crossvalidation_dataset_2; set 

Crossvalidation_dataset_1; by SNP&j; 

    if first.SNP&j;  

   run; 

   %let nobs=0; %nobs(Crossvalidation_dataset_2); 

 

   %do i=1 %to &nobs  ; 

    data Crossvalidation_dataset_3; set 

Crossvalidation_dataset_2; If _N_ eq &i; run; 

    proc sql noprint; 

     select SNP&j into :Cross_val_allele  from 

Crossvalidation_dataset_3; 

    quit; run; 

    Data Crossvalidation_allele; set 

Crossvalidation_dataset_1; if SNP&j eq &Cross_val_allele;  

    run; 

    %let nobs=0; %nobs(Crossvalidation_allele); 

    %let randomini=%sysevalf(&randomini + 10); 

 

    data Unif(keep= m); 

    call streaminit(%sysevalf(&randomini)); 

    a = -1; b = 1; 

    Min = 1; Max = 5; 

    do i = 1 to &NObs; 

       u = rand("Uniform");    /* U[0,1] */ 

       m = min + floor((1+Max-Min)*u); /* uniform integer 

in Min..Max */ 

       output; 

    end; 

    run; 

    Data Crossvalidation_allele; merge 

Crossvalidation_allele unif; if m ne 1; drop m; Select=1; run;  

    %if &i eq 1 %then %do; 

     Data CV_data; set Crossvalidation_allele; run; 

    %end; 

    %else %do; 

     Proc append Base=CV_data 

data=Crossvalidation_allele force ; run ;  

    %end; 

   %end; 

   Proc sort data=Crossvalidation_dataset; by genotype; run;  

   Proc sort data=CV_data; by genotype; run;  

 

   Data g_CV; merge Crossvalidation_dataset CV_data ; by 

genotype; if select eq 1; drop select; run; 

   proc delete data = work.Crossvalidation_dataset; run; 

   proc delete data = work.Crossvalidation_dataset_1; run; 

   proc delete data = work.Crossvalidation_dataset_2; run; 

   proc delete data = work.Crossvalidation_dataset_3; run; 

   proc delete data = work.Crossvalidation_allele; run; 

   proc delete data = work.Crossvalidation_allele_1; run; 

   proc delete data = work.Unif; run; 

   proc delete data = work.Cv_data; run; 

 

 



   /************** Model; change for your experimental design 

***/ 

 

 

   proc mixed data=G_cv;  

    class genotype SNP&j; 

    model y= SNP&j / htype=1 solution; 

    ods output   tests1=SNP_test_Anova; 

   run; 

 

 

  

 /*************************************************************/ 

 

 

   Data SNP_test_Anova; set SNP_test_Anova;  

    informat  effect $32.  Marker $25. probf 30.28; 

    format   effect $25.  Marker $25. probf 30.28; 

Marker="&Marker";  

   run; 

   %if &cv eq 1 %then %do; 

    Data Marker_Main_effect_CV; set SNP_test_Anova;  run; 

   %end; 

   %else %do; 

    Proc append Base=Marker_Main_effect_CV 

data=SNP_test_Anova force; run ;  

   %end; 

   proc delete data = work.SNP_test_Anova; run; 

  %end; 

  proc MEANS NOPRINT mean std 

   data=Marker_Main_effect_CV; 

   class Marker; 

   var Fvalue probF ; 

   output out=CVprobf   mean= ;  

   output out=cvprobfstd   stddev=;  

  run; 

  data cvprobF; set cvprobF;  if _type_ eq 1; CVfvalue=fvalue; 

CVprobf=probf; format Marker $25. CVprobf 30.28;  keep Marker CVfvalue CVprobf  

; run;  

  data cvprobfstd; set cvprobfstd; CVstdfvalue=fvalue; 

CVstdprobf=probf;  if _type_ eq 1; format Marker $25. CVstdprobf 30.28; keep 

Marker CVstdfvalue CVstdprobf; run; 

  data cvprobF; merge cvprobf cvprobfstd; run; 

  proc delete data = work.cvprobfstd; run; 

 

  %if &j eq 1 %then %do; 

   Data Marker_Main_cvprobF; set cvprobF;  run; 

  %end; 

  %else %do; 

   Proc append Base=Marker_Main_cvprobF data=cvprobF force;

 run ;  

  %end; 

  proc delete data = work.cvprobF; run; 

 

%Mend cross_validation; 

 

%macro marker_main_and_Cross_Val; 

 



 proc delete data = work.Marker_Main_effect; run; 

 

 %do j=1 %to   &No_SNP  ;  

 

  %let marker=marker_&j; 

 

  /*********** Hide log information **************************/ 

  %if &j gt 3  %then %do; 

    proc printto log=outlog;     run; 

    proc options;      run; 

    proc printto print = outdoc;  run;  

    ODS Results OFF;  

   %end; 

  /*************************************************************/ 

 

  /************** Model; change for your experimental design ***/ 

 

 

  proc mixed data=G;  

   class genotype SNP&j; 

   model y= SNP&j / htype=1 solution; 

   ods output   tests1=SNP_test_Anova; 

  run; 

 

 

  /*************************************************************/ 

 

 

  Data SNP_test_Anova; set SNP_test_Anova;  

   informat effect $32  Marker $25.  probf 30.28;  

   format   effect $32  Marker $25.  probf 30.28; 

Marker="&Marker";  

  run; 

 

  proc sql noprint; 

   select ProbF into :Cross_val_ProbF  from SNP_test_Anova; 

  quit; run; 

  

  %if &j eq 1 %then %do; 

   Data Marker_Main_effect; set SNP_test_Anova;  run; 

  %end; 

  %else %do; 

   Proc append Base=Marker_Main_effect data=SNP_test_Anova 

force; run ;  

  %end; 

 

  %if &Cross_val_ProbF le &CV_Prob %then %do; 

   %cross_validation; 

  %end; 

 

 %end; 

 %let nobs=0; %nobs(Marker_Main_cvprobF); 

 %if &nobs ge 1 %then %do; 

  Proc sort data=Marker_Main_effect; by Marker; run; 

  Proc sort data=Marker_Main_cvprobF; by Marker; run; 

  data Marker_Main_effect; merge Marker_Main_effect 

Marker_Main_cvprobF; by Marker; run; 

  proc delete data = work.Marker_Main_cvprobF; run; 



 %end; 

 

 Proc sort data=Marker_Main_effect; by ProbF; run; 

 proc delete data = work.SNP_test_Anova; run; 

 proc delete data = work.Marker_Main_effect_cv; run; 

 proc delete data = work.G_cv; run; 

 

 /*********** Calculation of FDR probability ******************/ 

 

 proc multtest inpvalues(ProbF)=Marker_Main_effect OUT=Marker_Main_effect 

FDR; run; 

 

 /*************************************************************/ 

 

 /**************** Print Log file information again ************/ 

 

 Proc printto; run;  

 

 /*************************************************************/ 

 

%Mend marker_main_and_Cross_Val; 

 

/*  Start the macro by calling it using %(the given) macro: 

 

Example for cross_validation in connection with Marker-trait associations 

data:  

 

Define probability of applying the Cross_Validation and number of runs: 

 

%let CV_Prob=0.05; %let No_CV_reps=20;  

 

%marker_main_and_Cross_Val; 

 

*/ 

 

 


